

MÒDUL 3

BIOLOGIA

Introducció

De forma descriptiva, tot i que incompleta, podríem definir la matemàtica com la ciència que estudia les propietats dels nombres, de les figures, dels conjunts, de les operacions, de les funcions, etc.

Però la matemàtica és quelcom més que tot això: és el llenguatge bàsic de totes les ciències. Es tracta d'una ciència transversal a través de la qual es modelen tots aquells sistemes físics que ens envolten, dia a dia.

Una bona mostra la trobem en les closques marines, les quals poden ser representades com a superfícies en tres dimensions (3D). La seva forma autosimilar es pot generar mitjançant una simple fórmula amb un cert nombre de paràmetres lliures.


A la natura, amb certa facilitat, també ens podem trobar amb determinades estructures que mantenen la seva forma essencial, fragmentada i irregular, tot i variant l'escala d'observació. D'aquestes estructures també se'n pot extreure un model matemàtic que ens permetrà, tant representar-les, com crear-ne de noves. Parlem dels fractals.

En aquest mòdul ens endinsem, doncs, en la comprensió de la grandesa de les matemàtiques com a ciència transversal, que ens permet modelitzacions tals com les formes de les closques marines i els fractals, així com en l'aprenentatge de les eines de representació gràfica de què disposa el Maple, que ens permet visualitzar i comprovar la similitud de la modelització matemàtica amb la realitat que ens envolta.

Closques marines

Introducció

Les closques marines es poden modelitzar mitjançant superfícies tridimensionals. La superfície de qualsevol closca marina pot ser generada mitjançant el moviment d'una corba generatriu que es mou al llarg d'una corba estructural. La corba estructural descriu la forma global de la closca, mentre que la corba generatriu modela la forma i obertura de la closca.




En aquest mòdul introduïrem un model depenent de 13 paràmetres per modelitzar les closques marines, basat en la referència [1].

El model complet el definirem en dues fases: primer definirem les equacions de la corba estructural i després definirem les equacions de la corba generatriu que ens permetran obtenir un model de les closques marines.

Corba estructural


Com ja hem dit, la estructura bàsica d'una closca marina ve definida per una corba en tres dimensions: la corba estructural. Aquesta corba, vista des de dalt (és a dir, la seva projecció sobre el pla XY), és una espiral logarítmica.




L'equació d'una espiral logarítmica és molt fàcil de donar en coordenades polars:

$$\rho = A \cdot e^{\theta \cdot \cot(\alpha)}$$

on el parell (ρ, θ) són les coordenades polars (radi i angle) i A i α són els paràmetres que determinen el tipus d'espiral logarítmica.



El paràmetre A varia l'obertura de l'espiral, mentre que el paràmetre α és l'angle que determinen la recta que passa per l'origen i els punts de la corba i la recta tangent de la corba en aquest punt.




En aquest cas és important observar que com es defineix l'angle θ des del punt fins a l'eix OX (que no és la forma estàndard). Llavors, un punt del pla (ρ, θ) en coordenades polars equival, en coordenades cartesianes, al punt

$$x = \rho \cdot \cos(\theta) = A \cdot e^{\theta \cdot \cot(\alpha)} \cdot \cos(\theta)$$

$$y = \rho \cdot (-\sin(\theta)) = -A \cdot e^{\theta \cdot \cot(\alpha)} \cdot \sin(\theta)$$

com es pot apreciar en la gràfica següent.



Per dibuixar una espiral logarítmica amb el Maple es pot fer amb les següents comandes:

```
> x:=A*exp(theta*cot(alpha))*cos(theta);
   y:=-A*exp(theta*cot(alpha))*sin(theta);
```

$$x := A e^{(\theta \cot(\alpha))} \cos(\theta)$$

$$y := -A e^{(\theta \cot(\alpha))} \sin(\theta)$$

```
> S:={A=6,alpha=4/9*Pi};
```


$$S := \left\{ \alpha = \frac{4}{9} \pi, A = 6 \right\}$$

```
> x:=subs(S,x);
   y:=subs(S,y);
```

$$x := 6 e^{\left(\theta \cot\left(\frac{4}{9} \pi\right)\right)} \cos(\theta)$$

$$y := -6 e^{\left(\theta \cot\left(\frac{4}{9} \pi\right)\right)} \sin(\theta)$$

```
> plot([x,y,theta=0..4*evalf(Pi)]);
```



Primer es defineixen les coordenades x i y en funció dels dos paràmetres A i α . Després donem valors als dos paràmetres dins d'una llista S que mitjançant la comanda subs (substituir) els substituïm a les equacions. Finalment dibuixem usant la comanda plot la corba en forma paramètrica on fem variar l'angle θ entre 0 i 4π . (NOTA: per a escriure dues línies dins de la mateixa línia de comandes cal prémer Majúscules + Enter).

El dibuix també es pot fer en forma més compacta amb les comandes:

```
> x:=A*exp(theta*cot(alpha))*cos(theta);
  y:=-A*exp(theta*cot(alpha))*sin(theta);
```


$$x := A e^{(\theta \cot(\alpha))} \cos(\theta)$$

$$y := -A e^{(\theta \cot(\alpha))} \sin(\theta)$$

```
> S:={A=6,alpha=4/9*Pi};
```


$$S := \left\{ \alpha = \frac{4}{9} \pi, A = 6 \right\}$$

```
> plot(subs(S,[x,y,theta=0..4*Pi]),thickness=3);
```




on a més s'ha incorporat un paràmetre (thickness) per variar el gruix del dibuix.

La forma de moltes closques de mar segueixen la trajectòria d'una espiral logarítmica, encara que no necessàriament inscrita en un pla. A l'eivar-se en l'espai el resultat és una corba que està sobre la superfície d'un con. Aquesta és la corba estructural de les closques de mar.



Evidentment, la corba estructural depèn del con sobre el qual es recolza l'espiral logarítmica. En aquest cas, el con vindrà donat per l'angle β representat a la figura.



El punt del con que està a distància R de l'origen verifica que:

$$z = -R \cdot \cos(\beta)$$

$$\rho = \sqrt{x^2 + y^2} = R \cdot \sin(\beta)$$

Una manera de fer que l'espiral logarítmica que teníem:

$$\rho = A \cdot e^{\theta \cdot \cot(\alpha)}$$

es recolzi sobre un con d'angle β és modificar el radi de l'espiral logarítmica prenent:

$$A = A \cdot \sin(\beta)$$

de manera que obtenim les equacions:

$$x = A \cdot \sin(\beta) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \cos(\theta)$$

$$y = -A \cdot \sin(\beta) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \sin(\theta)$$

$$z = -A \cdot e^{\theta \cdot \cot(\alpha)} \cdot \cos(\beta)$$

Per tant la corba estructural depèn dels tres paràmetres A , α i β .

Per dibuixar-la amb el Maple es pot fer amb la comanda `spacecurve` que permet dibuixar una corba en l'espai que estigui expressada en forma paramètrica (com tenim en aquest cas):

```
> x:=A*sin(beta)*exp(theta*cot(alpha))*cos(theta);
y:=-A*sin(beta)*exp(theta*cot(alpha))*sin(theta);
z:=-A*exp(theta*cot(alpha))*cos(beta);
```

$$x := A \sin(\beta) e^{(\theta \cot(\alpha))} \cos(\theta)$$

$$y := -A \sin(\beta) e^{(\theta \cot(\alpha))} \sin(\theta)$$

$$z := -A e^{(\theta \cot(\alpha))} \cos(\beta)$$

```
> S:={alpha=78*Pi/180,beta=40*Pi/180,A=30};
```

$$S := \left\{ \alpha = \frac{13}{30} \pi, \beta = \frac{2}{9} \pi, A = 30 \right\}$$

```
> spacecurve(subs(S,[x,y,z,theta=0..4*Pi]),thickness=3);
```

Amb el Maple és senzill dibuixar la corba estructural i el con de recolzament per comprovar que realment la corba està sobre el con. Per això haurem de fer dos dibuixos alhora, i ho farem amb la comanda `display`. Per poder utilitzar aquesta comanda cal carregar el paquet `plots`. La càrrega de paquets es fa mitjançant la comanda `with`, és a dir, la següent línia:

```
> with(plots):
```


ens permet usar la comanda `display`. Per veure les comandes associades al paquet `plots`, es pot posar la comanda

```
> with(plots);
```

acabada amb punt i coma de manera que obtindrem un llistat de les comandes accessibles a través d'aquest paquet.

El dibuix de les dues corbes es pot fer utilitzant les següents comandes:

```
> display(
  spacecurve(subs(S,[x,y,z,theta=0..4*Pi])
    ,thickness=3,color=black),
  plot3d(subs(S,
    [a*sin(beta)*cos(theta),a*sin(beta)*sin(theta),a*cos(beta)]
    ),a=-500..0,theta=0..2*Pi
    ,style=patchnogrid,transparency=0.7),orientation=[40,80]
);
```

Abans de passar a la definició de la corba generatriu, és interessant implementar la corba estructural en el Maple i veure les diferents formes que ens apareixen al variar els tres paràmetres que determinen la corba estructural. També és interessant entendre i jugar, mirant l'ajuda del Maple, amb les opcions de dibuix emprades en el dibuix de la corba estructural i el con.


Corba generatriu

La segona corba que ens permet modelitzar les closques marines és la corba generatriu. Aquesta ens permet generar la superfície exterior de la closca al moure's al llarg de la corba estructural.

En aquest model considerarem que la corba generatriu té forma el·lipsoidal. L'equació implícita d'una el·lipse centrada en el punt (0,0) és:

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$$

on a i b són els paràmetres de l'el·lipse que representen la meitat de les longituds dels eixos major i menor respectivament.



Si volem obtenir una representació en forma polar en funció de l'angle s , cal trobar una relació entre el radi ρ i l'angle s . En aquest cas si tenim en compte que:

$$x = \rho \cdot \cos(s)$$

$$y = \rho \cdot \sin(s)$$

i ho substituïm en l'equació de l'el·lipse tenim que:

$$\frac{(\rho \cdot \cos(s))^2}{a^2} + \frac{(\rho \cdot \sin(s))^2}{b^2} = 1$$

d'on treiem que:

$$\rho = \frac{1}{\sqrt{\frac{\cos(s)^2}{a^2} + \frac{\sin(s)^2}{b^2}}}$$


A partir d'ara denominarem per $h(s)$ la funció que donat un paràmetre s , ens dóna el radi de l'el·lipse centrada en el $(0,0)$. D'aquesta manera, l'el·lipse es pot parametritzar en coordenades cartesianes com:

$$x = h(s) \cdot \cos(s)$$

$$y = h(s) \cdot \sin(s)$$

$$\text{on } h(s) = \frac{1}{\sqrt{\frac{\cos(s)^2}{a^2} + \frac{\sin(s)^2}{b^2}}}$$

Observem que només amb els dos paràmetres a i b que ens donen les longituds dels semieixos de l'el·lipse, tenim que la corba generatriu és constant al llarg de la corba estructural.



Això no és el que passa a la realitat, on a mesura que ens anem allunyant del centre de la closca, els semieixos de la corba generatriu creixen. De fet, creixen amb la mateixa proporció que la corba estructural. Per això, de fet, el radi de la corba generatriu $h(s)$ apareix multiplicat pel factor $\rho = e^{\theta \cdot \cot(\alpha)}$

D'aquesta manera tenim la parametrització de l'el·lipse:

$$x = h(s) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \cos(s)$$

$$y = h(s) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \sin(s)$$

L'únic que cal en aquest punt per obtenir el model més simple de closca marina que veurem és a cada punt de la corba estructural sumar-li les coordenades anteriors (és a dir, centrem la corba generatriu en la corba estructural):

$$\begin{aligned} x &= A \cdot \sin(\beta) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \cos(\theta) + h(s) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \cos(s) \\ &= (A \cdot \sin(\beta) \cdot \cos(\theta) + h(s) \cdot \cos(s)) \cdot e^{\theta \cdot \cot(\alpha)} \end{aligned}$$

$$\begin{aligned} y &= -A \cdot \sin(\beta) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \sin(\theta) + h(s) \cdot e^{\theta \cdot \cot(\alpha)} \cdot \sin(s) \\ &= (-A \cdot \sin(\beta) \cdot \sin(\theta) + h(s) \cdot \sin(s)) \cdot e^{\theta \cdot \cot(\alpha)} \end{aligned}$$

$$z = -A \cdot e^{\theta \cdot \cot(\alpha)} \cdot \cos(\beta)$$

En aquest punt el nostre model depèn de 5 paràmetres (3 de la corba estructural i 2 de la corba generatriu). La gràfica de les closques que podem obtenir amb aquesta modelització es poden realitzar usant les següents comandes:

> **h:=s->1/sqrt(cos(s)^2/a^2+sin(s)^2/b^2);**

$$h := s \rightarrow \frac{1}{\sqrt{\frac{\cos(s)^2}{a^2} + \frac{\sin(s)^2}{b^2}}}$$

> **x:=(A*sin(beta)*cos(theta)+h(s)*cos(s))**
***exp(theta*cot(alpha));**

y:=(-A*sin(beta)*sin(theta)+h(s)*sin(s))
***exp(theta*cot(alpha));**

z:=-A*cos(beta)*exp(theta*cot(alpha));

$$x := \left(A \sin(\beta) \cos(\theta) + \frac{\cos(s)}{\sqrt{\frac{\cos(s)^2}{a^2} + \frac{\sin(s)^2}{b^2}}} \right) e^{(\theta \cot(\alpha))}$$

$$y := \left(-A \sin(\beta) \sin(\theta) + \frac{\sin(s)}{\sqrt{\frac{\cos(s)^2}{a^2} + \frac{\sin(s)^2}{b^2}}} \right) e^{(\theta \cot(\alpha))}$$



$$z := -A e^{(\theta \cot(\alpha))} \cos(\beta)$$

```
> S:={alpha=80*Pi/180,beta=40*Pi/180,A=25,a=12,b=16};
```

$$S := \left\{ \beta = \frac{2}{9} \pi, \alpha = \frac{4}{9} \pi, a = 12, A = 25, b = 16 \right\}$$

```
> plot3d(subs(S,[x,y,z]),theta=-2*Pi..4*Pi,s=-Pi..Pi,
scaling=constrained,orientation=[-68,80],grid=[100,100]);
```

La parametrització anterior dóna lloc a les següents gràfiques:



que queden lluny de la modelització que estàvem buscant. Això és degut a que hem suposat que la corba generatriu es troba en tot moment en un pla paral·lel al pla XY com es veu en la figura de l'esquerra. Per modelitzar correctament les closques marines, haurem de moure la corba generatriu en un pla que no sigui paral·lel al pla XY (introduint 3 nous paràmetres).

Abans de variar el pla on descriurem la corba generatriu, introduïrem una nova notació que ens facilitarà la comprensió del model que volem plantejar. Per facilitar l'escriptura separarem les coordenades de la corba estructural (x_e, y_e, z_e) de les de la corba generatriu (x_g, y_g, z_g) que sumades i multiplicades per $e^{\theta \cdot \cot(\alpha)}$ donaran lloc a les coordenades dels punts de la closca. És a dir:

$$x = (x_e + x_g) \cdot e^{\theta \cdot \cot(\alpha)}$$

$$y = (y_e + y_g) \cdot e^{\theta \cdot \cot(\alpha)}$$

$$z = (z_e + z_g) \cdot e^{\theta \cdot \cot(\alpha)}$$

on:

$$x_e = A \cdot \sin(\beta) \cdot \cos(\theta)$$

$$y_e = -A \cdot \sin(\beta) \cdot \sin(\theta)$$


$$z_e = -A \cdot \cos(\beta)$$

$$x_g = h(s) \cdot \cos(s)$$

$$y_g = h(s) \cdot \sin(s)$$

$$z_g = 0$$

El primer que realitzarem serà una rotació (centrada en el punt (x_e, y_e, z_e)) respecte l'eix major de l'el·lipse (eix OX) d'angle μ .





Aquesta rotació ve donada per la matriu T1:


$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos(\mu) & \sin(\mu) \\ 0 & -\sin(\mu) & \cos(\mu) \end{bmatrix}$$

Per tant, les noves coordenades de la corba generatriu venen donades per:

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos(\mu) & \sin(\mu) \\ 0 & -\sin(\mu) & \cos(\mu) \end{bmatrix} \begin{bmatrix} xg \\ yg \\ zg \end{bmatrix} = \begin{bmatrix} xg \\ yg \cdot \cos(\mu) \\ -yg \cdot \sin(\mu) \end{bmatrix} = \begin{bmatrix} h(s)\cos(s) \\ h(s)\sin(s)\cos(\mu) \\ -h(s)\sin(s)\sin(\mu) \end{bmatrix}$$



Tot i això, amb aquest gir tampoc aconseguim la modelització que buscàvem. El motiu es veu en la següent figura



on es veu clarament que la corba generatriu no es troba situada en el pla XY sinó en el pla vertical que passa pel centre (0,0,0) i pel punt en què estem (x,y,z). Per tant, hem de fer un canvi d'eixos de coordenades (rotació) sobre l'expressió que tenim de la corba generatriu.

Aquesta rotació ve donada per la matriu de canvi T2:

$$\begin{bmatrix} \cos(\theta) & 0 & \sin(\theta) \\ -\sin(\theta) & 0 & \cos(\theta) \\ 0 & 1 & 0 \end{bmatrix}$$


D'aquesta manera obtenim:

$$\begin{bmatrix} \cos(\theta) & 0 & \sin(\theta) \\ -\sin(\theta) & 0 & \cos(\theta) \\ 0 & 1 & 0 \end{bmatrix} \begin{bmatrix} h(s)\cos(s) \\ h(s)\sin(s)\cos(\mu) \\ -h(s)\sin(s)\sin(\mu) \end{bmatrix} = \begin{bmatrix} h(s)\cos(s)\cos(\theta) - h(s)\sin(s)\sin(\mu)\sin(\theta) \\ -h(s)\cos(s)\sin(\theta) - h(s)\sin(s)\sin(\mu)\cos(\theta) \\ h(s)\sin(s)\cos(\mu) \end{bmatrix}$$



En aquest cas veiem que si fem el dibuix s'aproxima molt més a la superfície d'una closca marina:

```
> xg:=h(s)*(cos(s)*cos(theta)-sin(s)*sin(mu)*sin(theta)):
  yg:=-h(s)*(cos(s)*sin(theta)+sin(s)*sin(mu)*cos(theta)):
  zg:=h(s)*sin(s)*cos(mu):
> x:=(xe+xg)*exp(theta*cot(alpha)):
  y:=(ye+yg)*exp(theta*cot(alpha)):
  z:=(ze+zg)*exp(theta*cot(alpha)):
> S:={alpha=4*Pi/9,beta=2*Pi/9,A=25,a=12,b=16,mu=Pi/90}:
```

```
> plot3d(subs(S,[x,y,z]),theta=-2*Pi..4*Pi,s=-Pi..Pi,  
scaling=constrained,orientation=[0,90],grid=[100,100]);
```



Finalment, per obtenir el model més senzill de closca marina, es realitzen dues rotacions més descrites en les dues figures següents:



La primera rotació ens diu que el pla sobre el qual es troba la generatriu no es el pla radial sinó que està rotat un angle Ω . Aquesta rotació ve donada per la matriu de transformació T2:

$$\begin{bmatrix} \cos(\Omega) & \sin(\Omega) & 0 \\ -\sin(\Omega) & \cos(\Omega) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

El segon gir és un gir respecte un eix normal a la generatriu d'angle Φ i és molt senzill de realitzar ja que simplement cal sumar Φ a l'angle s .

Llavors fent aquestes dues transformacions les coordenades de la corba generatriu s'obtenen fent el següent producte:

$$\begin{bmatrix} \cos(\Omega) & \sin(\Omega) & 0 \\ -\sin(\Omega) & \cos(\Omega) & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} h(s)\cos(s + \Phi)\cos(\theta) - h(s)\sin(s + \Phi)\sin(\mu)\sin(\theta) \\ -h(s)\cos(s + \Phi)\sin(\theta) - h(s)\sin(s + \Phi)\sin(\mu)\cos(\theta) \\ h(s)\sin(s + \Phi)\cos(\mu) \end{bmatrix}$$

Realitzant el producte i utilitzant les fórmules trigonomètriques següents:

$$\cos(\theta + \Omega) = \cos(\theta) \cos(\Omega) - \sin(\theta) \sin(\Omega)$$

$$\sin(\theta + \Omega) = \sin(\theta) \cos(\Omega) + \cos(\theta) \sin(\Omega)$$

s'arriba a les equacions per a la corba generatriu següents:



$$xg = h(s) (\cos(\theta + \Omega) \cos(s + \Phi) - \sin(\theta + \Omega) \sin(\mu) \sin(s + \Phi))$$

$$yg = -h(s) (\sin(\theta + \Omega) \cos(s + \Phi) + \cos(\theta + \Omega) \sin(\mu) \sin(s + \Phi))$$

$$zg = h(s) \sin(s + \Phi) \cos(\mu)$$

El model resultant es pot descriure amb Maple segons:

```
> xe:=A*sin(beta)*cos(theta):
  ye:=-A*sin(beta)*sin(theta):
  ze:=-A*cos(beta):
> xg:=h(s)*(-sin(theta+Omega)*sin(mu)*sin(s+Phi)
  +cos(theta+Omega)*cos(s+Phi)):
  yg:=-h(s)*(sin(theta+Omega)*cos(s+Phi)
  +cos(theta+Omega)*sin(mu)*sin(s+Phi)):
  zg:=h(s)*sin(s+Phi)*cos(mu):
> h:=s->1/sqrt(cos(s)^2/a^2+sin(s)^2/b^2):
> x:=(xe+xg)*exp(theta*cot(alpha)):
  y:=(ye+yg)*exp(theta*cot(alpha)):
  z:=(ze+zg)*exp(theta*cot(alpha)):
> S:={alpha=8*Pi/18,beta=4*Pi/18,
  A=25,a=12,b=16,
  mu=Pi/18,Omega=3*Pi/18,Phi=55*Pi/180}:
> plot3d(subs(S,[x,y,z]),theta=-2*Pi..4*Pi,s=-Pi..Pi,
  scaling=constrained,orientation=[-68,80],grid=[100,100]);
```





Model complet

Si considerem el model de 8 paràmetres (tres de la corba estructural A , α i β , dos de la corba generatriu a i b i tres més de rotació del pla on es troba la corba generatriu μ , Ω i Φ) que hem estudiat fins al moment, veiem que la corba generatriu és una corba el·lipsoidal on el radi ve donat per

$$h(s) \cdot e^{\theta \cdot \cot(\alpha)}$$

La funció $h(s)$ ens dóna el radi que té una el·lipse de semieixos major i menor a i b respectivament, mentre que el factor que conté l'exponencial fa que l'el·lipse augmenti a mesura que la closca marina va creixent.

De totes maneres, amb aquest model no es poden obtenir closques marines de l'estil:



ja que en aquests casos les seccions no són el·lipses sinó que contenen protuberàncies. Per modelitzar-les el que es fa és introduir cinc nous paràmetres que ens modificaran la definició de la funció $h(s)$.

El model que considerarem defineix:

$$h(s, \theta) = h(s) + Le^{-\left(\frac{2(s-P)}{W1}\right)^2 - \left(2\frac{l(\theta)}{W2}\right)^2}$$

on

$$h(s) = \frac{1}{\sqrt{\frac{\cos(s)^2}{a^2} + \frac{\sin(s)^2}{b^2}}}$$

i




$$l(\theta) = \frac{2\pi}{N} \left(\frac{N\theta}{2\pi} - \text{arrodoniment}_{\text{enter}} \left(\frac{N\theta}{2\pi} \right) \right)$$

NOTA: en Maple per arrodonir un nombre a l'enter més proper es pot usar la comanda round.

Els 5 nous paràmetres introduïts són: P , L , $W1$, $W2$ i N i venen definits per:

- P : angle que descriu la posició de les protuberàncies al llarg de l'el·lipse
- L : mesura l'alçada de cada protuberància
- $W1$: mesura l'amplada de cada protuberància al llarg de l'el·lipse
- $W2$: mesura l'amplada de cada protuberància al llarg de la corba estructural
- N : nombre de protuberàncies que es troben per cada gir de 2π de l'angle θ


Els diferents paràmetres es poden veure en les següents figures:



Exemples


En aquesta secció es detallen els paràmetres necessaris per reproduir les closques marines següents:

NATALINA



$$S := \{A = 25, a = 12, b = 16, P = 0, L = 0, \alpha = \frac{4\pi}{9}, \beta = \frac{2\pi}{9}, \mu = \frac{\pi}{18}, \Omega = \frac{\pi}{6}, \Phi = \frac{11\pi}{36}\}$$

NAUTILUS




$$S := \{b = 1.5, L = 0, A = 2, a = 2, P = 0, \beta = \frac{\pi}{2}, \phi = 0, \mu = 0, \alpha = \frac{4\pi}{9}, \Omega = 0\}$$

LYRIA


$$S := \left\{ P=0, A=50, a=40, b=14, L=8, N=8, \beta = -\frac{19\pi}{180}, \phi = \frac{\pi}{4}, \mu = \frac{\pi}{180}, \Omega = -\frac{\pi}{90}, \right. \\ \left. W_1 = \frac{\pi}{30}, W_2 = \frac{3\pi}{20}, \alpha = \frac{7\pi}{15} \right\}$$

TURRITELLA


$$S := \left\{ \beta = \frac{\pi}{45}, \alpha = \frac{89\pi}{180}, \phi = \frac{11\pi}{36}, A=22.2, a=1.3, b=1.5, L=0, P=0, \mu = \frac{\pi}{180}, \Omega = -\frac{\pi}{90} \right\}$$

OXYTELE


$$S := \left\{ L=0, W_1=6, W_2=27, b=19, A=47, \alpha = \frac{17\pi}{36}, \beta = \frac{7\pi}{180}, \phi = -\frac{\pi}{5}, P=0, a=40, N=8, \right. \\ \left. \mu = \frac{\pi}{180}, \Omega = -\frac{\pi}{90} \right\}$$

PLANORBIS


$$S := \left\{ \beta = \frac{17\pi}{36}, A = 45, a = 20, b = 30, L = 0, \phi = \frac{\pi}{18}, W_1 = 6, W_2 = 27, \Omega = \frac{\pi}{36}, \mu = \frac{\pi}{4}, P = 0, \right. \\ \left. N = 8, \alpha = \frac{7\pi}{15} \right\}$$

AMMONITE


$$S := \left\{ \alpha = \frac{83\pi}{180}, \phi = \frac{\pi}{180}, A = 2.5, a = 1, b = 0.9, P = 10, W_2 = 20, W_1 = 100, N = 15, L = \frac{1}{2}, \right. \\ \left. \Omega = \frac{\pi}{180}, \beta = \frac{\pi}{2}, \mu = \frac{\pi}{180} \right\}$$

CONUS

$$S := \left\{ \alpha = \frac{29\pi}{60}, \phi = \frac{13\pi}{30}, L = 0, W_1 = 6, W_2 = 27, \beta = \frac{7\pi}{180}, P = 0, \mu = 0, \Omega = 0, A = 7, a = 4.3, \right. \\ \left. N = 8, b = 1 \right\}$$

NAUTILUS

$$S := \left\{ L = 0, W_1 = 6, W_2 = 27, A = 450, a = 400, b = 310, \Omega = -\frac{2\pi}{9}, \alpha = \frac{\pi}{10}, P = 0, \beta = \frac{\pi}{2}, \phi = 0, \right. \\ \left. \mu = 0, N = 8 \right\}$$

HELCION


$$S := \left\{ b = 1.5, L = 0, A = 2, a = 2, P = 0, \beta = \frac{\pi}{2}, \phi = 0, \mu = 0, \alpha = \frac{4\pi}{9}, \Omega = 0 \right\}$$

CODAKIA

$$S := \{L = 0, \alpha = \frac{37\pi}{180}, W_1 = 6, W_2 = 27, \Omega = \frac{\pi}{180}, P = 0, \beta = \frac{\pi}{2}, \phi = 0, N = 8, \mu = \frac{\pi}{180}, \\ A = 10500, a = 10000, b = 10500\}$$

CLOSCA AMB PROTUBERÀNCIES

$$SI := \{\alpha = \frac{83\pi}{180}, \phi = \frac{\pi}{180}, A = 2.5, a = 1, b = 0.9, \Omega = \frac{\pi}{180}, \beta = \frac{\pi}{2}, \mu = \frac{\pi}{180}\}$$

STRUTHIOLARIA

$$S := \{\alpha = \frac{29\pi}{60}, L = \frac{1}{2}, A = 5, a = 1.8, b = 1.6, N = 16, P = -\frac{\pi}{12}, \beta = \frac{13\pi}{180}, \phi = \frac{2\pi}{9}, \mu = \frac{\pi}{18}, \\ \Omega = \frac{\pi}{6}, W_1 = \frac{\pi}{36}, W_2 = \frac{\pi}{36}\}$$

És un bon exercici reproduir les closques anteriors i jugar amb les opcions de dibuix (style, grid, light, shading, orientation) mitjançant el menú interactiu de dibuix i també dins les línies de comandes. És important també, a l'hora d'ensenyar bé els gràfics, saber triar l'orientació des d'on es veu el dibuix.

Referència bàsica

A. Cortie : *Digital Seashells Comput & Graphics* Vol 17, pp. 79-84 (1993)

Fractals

Introducció

La geometria tradicional o euclídia, és la branca de les matemàtiques que s'encarrega de les propietats i de les mesures d'elements com ara punts, línies, plans i volums. La geometria euclídia també descriu els conjunts formats per la reunió dels elements que ja hem citat, les combinacions dels quals poden formar figures o formes específiques.

No obstant, les formes que es poden trobar a la natura (muntanyes, costes, núvols, fulles, arbres, flocs de neu, ...) no es poden descriure fàcilment per la geometria tradicional.





En aquest sentit, la **geometria fractal** proporciona una descripció i un model matemàtic per a les aparentment complicades formes de la natura. Aquestes formes posseeixen, a vegades, una remarcable invariància de simplificació sota els canvis de la magnificació, propietat que caracteritza als fractals.

Què és un fractal?

El fractal és, matemàticament, una figura geomètrica que és complexa i detallada en estructura a qualsevol nivell de magnificació. Sovint, els fractals són semblants a si mateixos, és a dir, posseeixen la propietat de que cada petita porció del fractal pot ser visualitzada com una rèplica a escala reduïda.

És important reconèixer que els fractals són una idealització. Cap corba en el món real és un veritable fractal; els objectes reals són produïts per processos que actuen només sobre un rang d'escala finites. En altres paraules, els objectes reals no tenen la infinita quantitat de detalls que els fractals ofereixen amb un cert grau de magnificació.

Es podria dir que un fractal és un patró infinit comprimit en un espai finit. Un bon exemple es veu fent un zoom en el conjunt de Mandelbrot que explicarem més endavant.



Pinzellada històrica

Els fractals es van concebre aproximadament el 1890 pel francès Henri Poincaré. Les seves idees van ser desenvolupades posteriorment pels matemàtics francesos Gaston Julia i Pierre Fatou, cap al 1918. Es va treballar amb intensitat durant aquests anys, però l'estudi va quedar congelat als anys 20.


L'estudi dels fractals es va reprendre a partir de 1974 a la IBM i va ser impulsat amb força gràcies a les primeres computadores digitals. El Dr. Mandelbrot, de la Universitat de Yale, és considerat el pare de la geometria fractal. En el seu honor, un dels conjunts que va investigar va rebre el seu nom.

Tipus de fractals

Els fractals poden ser generats en funció de diversos mètodes. D'entre aquests mètodes destaquem:

- **Algoritmes d'escapament.** En aquest tipus de fractals, considerem cada un dels punts del pla complex i s'aplica una fórmula de forma iterativa

fins que se satisfà una certa condició. En aquest moment, s'assigna un color al punt (nombre complex), en funció del nombre de repeticions fins que s'ha satisfet la condició. Fractals d'aquest tipus requereixen un gran nombre d'operacions i, per tant, només poden ser representats amb l'ajuda d'un ordinador. D'aquest tipus són el *conjunt de Mandelbrot* i el *conjunt de Julia*.




- **Funcions iterades.** El sistema de funcions iterades (IFS) és un mètode creat per M. Barnsley, basat en el principi d'auto semblança. En un fractal IFS, sempre és possible trobar una part de la figura que guarda una relació de semblança amb la figura sencera, tot i que a vegades és difícil d'apreciar. D'aquest tipus és l'anomenada *fulla fractal* (*fractal leaf*).



- **Fractals atractors.** Aquest tipus de fractals són la visualització del mètode de Newton per a calcular les solucions complexes de les equacions $z^n - 1 = 0$, a través de la fórmula iterativa $z_{j+1} = z_j - \frac{f_n(z_j)}{f_n'(z_j)}$, on $f_n(z) = z^n - 1$ i $n > 2$. El gràfic mostra els punts inicials pels quals el mètode és o no convergent cap a alguna de les arrels.
- **Fractals 'a mà'. Triangle de Sierpinski.** El matemàtic polac W. Sierpinski va introduir aquest fractal el 1919. Es genera considerant inicialment ($n=0$) un triangle equilàter de costat unitat. Seguidament ($n=1$), si agafem els punts mitjos de cada costat, podem construir un altre triangle equilàter invertit de costat $\frac{1}{2}$. Ara, repetim el procés ($n=2$), amb els 3 triangles de costat $\frac{1}{2}$ que estan orientats com l'inicial. Si repetim infinitament aquest procés, obtindrem una figura fractal que rep el nom

de triangle de Sierpinski. De forma similar, es pot generar el floc de neu de Koch.



El conjunt de Mandelbrot

El conjunt de Mandelbrot és un conjunt sorprenentment fàcil d'explicar, malgrat la seva complicació gràfica. Es basa en iteracions de la funció complexa

$$f: \mathbb{C} \rightarrow \mathbb{C}, f(z) = z^2 + c$$

on c és un nombre complex fix i z és sempre inicialment zero.

Per a cada valor de $c=x+yi$, la successió dels termes $f(0)=c$, $f(f(0))=f(c)=c^2+c$, $f(f(f(0)))=f(c^2+c)=(c^2+c)^2+c, \dots$, $(f^{@@n})(0)^1, \dots$ pot ser convergent o divergent. Donat que no podem iterar aquesta funció infinitament, considerarem que, donat un complex c , la successió és divergent si algun element de la successió té mòdul igual o superior a 2. D'altra banda, si cap dels primers n termes (en el codi hem considerat un màxim de 30 iteracions) de la successió té mòdul superior a 2, considerarem que la successió és convergent.

En la taula següent, hem considerat dos punts inicials $c_1 = i$ i $c_2 = 1 - i$. Cada un d'aquests punts, en ser iterats, ens resulta en dues successions

$c_1 = i$		$c_2 = 1 - i$	
$f(0) = c_1 = i$	mòdul = 1	$f(0) = c_2 = 1 - i$	mòdul = $\sqrt{2}$
$f(f(0)) = -1 + i$	mòdul = $\sqrt{2}$	$f(f(0)) = 1 - 3i$	mòdul = $\sqrt{10} > 2$

¹ Amb $(f^{@@n})$ volem expressar la composició de la funció f amb si mateixa n vegades.

$f(f(f(0)))=-i$	mòdul = 1	$f(f(f(0)))=-7-7i$	mòdul = $7\sqrt{2}$
$(f@@4)(0)=-1+i$	mòdul = $\sqrt{2}$	$(f@@4)(0)=1+97i$	mòdul = $\sqrt{9410}$
$(f@@5)(0)=-i$	mòdul = 1	$(f@@5)(0)=-9407+193i$	mòdul = $\sqrt{88528898}$
...
$(f@@29)(0)=-1+i$	mòdul = $\sqrt{2}$
$(f@@30)(0)=-i$	mòdul = 1
$m=30$		$m=2$	

Podem observar com, clarament, en el primer cas no divergim a l'infinit, mentre que en el segon cas, quan fem la segona iteració, el mòdul del nombre complex ja és superior a 2. D'aquesta manera, pintarem el complex $c_1 = i$ amb el color 30, mentre que el complex $c_2 = 1 - i$ serà pintat de color 2. Farem aquesta operació amb cada un dels punts.


Finalment, definim el conjunt de Mandelbrot com el conjunt de nombres complexos c que, en ser iterats com acabem de veure, no divergeixen. En el nostre gràfic, és la zona tancada de color vermell.

Però, com podem dibuixar aquest conjunt amb el Maple? De fet, no només representarem el conjunt, sinó també els punts de fora del conjunt, amb un color diferent en funció del nombre d'iteracions. Així doncs, podem definir una procediment (que farem servir per donar color al conjunt de Mandelbrot) de la següent manera:

```
> mandelbrot := proc(x, y)
  local c, z, m;
  c := x+y*I;
  z := 0;
  for m to 30 while abs(z) < 2 do
    z := z^2+c;
  od;
  m;
end;
```

La rutina **mandelbrot** retorna el valor m . Com més gran és m , més gran és la probabilitat que el punt $x+yi$ sigui del conjunt de Mandelbrot. En aquest cas, hem escollit un màxim de $m=30$, tot i que és possible que algun dels punts amb $m=30$ no siguin realment del conjunt (però és que no podem fer infinites iteracions!). Aquest nombre m es fa servir en l'opció de visualització 'color=mandelbrot' (veure a l'ajuda del Maple plot3d/option).

```
> plot3d(0, -2 .. 0.7, -1.2 .. 1.2, orientation=[-90,0],
  grid=[250, 250], style=patchnograd,
  scaling=constrained, color=mandelbrot);
```



Fixeu-vos, no obstant, que el Maple associa el mateix color al 0 que al 30!

És un bon exercici que modifiqueu el valor màxim que pot prendre la sortida m del procediment mandelbrot, així com que modifiqueu la sortida posant $\ln(m)$, $\sin(n)$, etc. També podeu modificar el segon i tercer argument del `plot3d` (eixos), per observar alguna zona concreta del conjunt.

El conjunt de Julia

El conjunt de Julia es defineix fent servir la mateixa funció que en el conjunt de Mandelbrot:

$$z \rightarrow z^2 + c.$$

Ara però, cada conjunt de Julia tindrà **associat** un mateix nombre complex c . Per a cada nombre complex del pla $z=x+yi$, la successió que iterem ara és:

$$\begin{aligned} f(z) &= z^2 + c, \\ f(f(z)) &= f(z^2 + c) = (z^2 + c)^2 + c, \\ f(f(f(z))) &= f((z^2 + c)^2 + c) = ((z^2 + c)^2 + c)^2 + c, \dots \end{aligned}$$

Noteu la diferència d'aquesta successió en relació a la que origina el conjunt de Mandelbrot.

Aleshores, donat un nombre complex c , el conjunt de Julia associat a aquest nombre complex és el conjunt dels nombres complexos $z_0=x+yi$ tals que en aplicar la funció de forma iterativa, el resultat no divergeix: és a dir, z_0 pertany al conjunt de Julia associat al nombre complex c si el mòdul del nombre complex

$$((z_0^2 + c)^2 + c) \dots$$


no divergeix. A efectes de computació considerarem, com el en cas anterior, que divergim si en algun moment el mòdul d'algun dels elements de la successió és més gran o igual a 2.

El procediment Maple que generem és:


```
> JuliaSet := proc(a, b)
  local z1, z2, z1s, z2s, m;
  (z1, z2) := (a, b);
  z1s := z1^2: z2s := z2^2;
  for m to 30 while z1s + z2s < 4 do
    (z1, z2) := (z1s - z2s, 2*z1*z2) + c:
    z1s := z1^2: z2s := z2^2;
  end do;
  m;
end:
```

Fixem-nos que en aquest cas, en lloc de la notació complexa, hem fet servir la notació que representa un nombre complex com a un parell ordenat, per millorar l'eficiència computacional.


```
> c := 0, 0.75;
> plot3d(0, -1.5 .. 1.5, -1.4 .. 1.4, orientation=[-90,0],
  grid=[250, 250], style=patchnogrid,
  scaling=constrained, color=JuliaSet);
```



A dalt hem representat el conjunt de Julia associat al nombre complex $0,75i$.



$$c = -0.8911 + 0.2578i$$



$$c = -0.194 + 0.6557i$$

Fractals atractors

Considerem per exemple, l'equació $z^3 - 1 = 0$, que té tres arrels complexes (una real i dues complexes conjugades):

> `solve(z^3-1, z);`

$$1, -\frac{1}{2} + \frac{1}{2}i\sqrt{3}, -\frac{1}{2} - \frac{1}{2}i\sqrt{3}$$

Donat un punt del pla complex qualsevol, sabem que, en aplicar el mètode de Newton de forma iterativa aquest punt, convergirem a alguna d'aquestes tres arrels o divergirem. Per exemple, si considerem el punt inicial $z_0 = 1 + i$, el resultat d'aplicar iterativament el mètode de Newton ens dóna:

$z_{j+1} = z_j - \frac{f_n(z_j)}{f_n'(z_j)} = z_j - \frac{z^3 - 1}{3z^2}$	
$z_0 = 1 + i$	$z_1 = \frac{2}{3} + \frac{1}{2}i$
$z_1 = \frac{2}{3} + \frac{1}{2}i$	$z_2 = -0.58 - 0.13i$
$z_2 = -0.58 - 0.13i$	$z_3 = 1.25 + 0.31i$
$z_3 = 1.25 + 0.31i$	$z_4 = 1.01 + 0.11i$
$z_4 = 1.01 + 0.11i$	$z_5 = 0.99 + 0.004i$
$z_5 = 0.99 + 0.004i$	$z_6 = 1$

Hem vist que en aquest cas hem convergit en 6 passos a una de les arrels (l'arrel real).

Si en canvi, com a punt inicial considerem $z_0 = -2 + i$, la seqüència d'aproximacions és:

$z_{j+1} = z_j - \frac{f_n(z_j)}{f_n'(z_j)} = z_j - \frac{z^3 - 1}{3z^2}$	
$z_0 = -2 + i$	$z_1 = -1.29 + 0.72i$
$z_1 = -1.29 + 0.72i$	$z_2 = -0.78 + 0.61i$
$z_2 = -0.78 + 0.61i$	$z_3 = -0.44 + 0.74i$
$z_3 = -0.44 + 0.74i$	$z_4 = -0.51 + 0.89i$
$z_4 = -0.51 + 0.89i$	$z_5 = -0.50 + 0.87i$
$z_5 = -0.50 + 0.87i$	$z_6 = -\frac{1}{2} + \frac{\sqrt{3}}{2}i$

En aquest segon punt, hem convergit cap a una de les dues arrels complexes conjugades, i també ho hem fet en 6 passos.

En el codi que implementa aquest tipus de fractals, hem considerat que hem arribat a convergir quan l'error absolut entre la solució real i la solució aproximada és inferior a 0.001. D'altra banda, iterem el procés fins a un màxim de 50 vegades.


El codi serà:

```
> newton := proc(x, y)
  local z, m;

  z := evalf(x+y*I);
  for m from 0 to 50 while abs(z^3-1) >= 0.001 do
    z := z - (z^3-1)/(3*z^2)
  od;


  m
end:

> plot3d(0, -2 .. 2, -1.5 .. 1.5, orientation=[-90,0],
  grid=[250,250], style=patchnograd,
  scaling=constrained, color=newton);
```




Podríeu localitzar en aquest gràfic les tres arrels del polinomi $p(z) = z^3 - 1$?


El fractal atractor de l'equació de cinquè grau $z^5 - 1 = 0$, que s'aconsegueix de la mateixa manera, és:



Podries fer ara un fractal atractiu que assignés un color a cada punt en funció de la seva tendència? És a dir, si un punt tendeix a la primera arrel, assignem el color a , si tendeix a la segona, assignem el color b , etc. El resultat obtingut ha de ser un dibuix com el mostrat a continuació.



N=3



N=5

Referències

<http://www.math.utsa.edu/mirrors/maple/mfrintro.htm>

<http://www.quanta.net.py/zfractal/intro.htm>